

Backtracking - n queens

OPTION INFORMATIQUE - TP n° 3.1 - Olivier Reynet

À la fin de ce chapitre, je sais :

- ☞ Implémenter un algorithme de recherche par la force brute
- ☞ Implémenter un algorithme de recherche par retour sur trace
- ☞ garantir qu'une fonction retourne toujours le même type et éventuellement `unit ()`.
- ☞ coder une fonction sur une liste de manière récursive en utilisant le filtrage de motif et les fonctions auxiliaires
- ☞ utiliser module `List` pour coder des équivalents aux codes récursifs (`iter`, `map`, `filter`, `fold`)

Ce TP a pour but d'appréhender la thématique de l'exploration par la force brute puis par la technique du retour sur trace.

A Le problème des n reines

On cherche à placer sur un échiquier de $n \times n$ cases n reines sans que celles-ci s'attaquent les unes les autres. On rappelle que la reine peut attaquer une pièce sur la ligne, la colonne et les diagonales qui partent de sa position.

B Modélisation de l'échiquier pour les n reines

Il est possible de représenter un échiquier à l'aide de différentes structures de données. La première qui vient à l'esprit, certainement à cause de la visualisation de l'échiquier, est le tableau à deux dimensions. Néanmoins, lorsqu'on observe de plus près la répartition des reines pour des configurations solutions, il apparaît clairement qu'une même ligne ne peut accueillir qu'une seule reine. C'est pourquoi il est possible d'implémenter l'échiquier par une liste : l'indice de la liste représente le numéro de la ligne sur laquelle se situe la reine. Le i ème élément de la liste représente la colonne sur laquelle se trouve la reine. S'il n'y a pas de reine sur la ligne i , le i ème élément de la liste vaut -1 .

Par exemple, l'échiquier représenté sur la figure 1 est encodé par la liste `board=[7;3;0;2;5;1;6;4]`. On note que le premier élément de `board` vaut 7, ce qui signifie qu'il y a une reine sur la première ligne et la huitième colonne. Un échiquier `[3;-1;2;-1]` est un échiquier de 4x4 avec une reine en (0,3) et une autre en 2,2.




FIGURE 1 – Exemple d'échiquier 8x8 solution du problème des huit reines.

C Résolution par force brute

C1. Combien y-a-t-il de solutions au problème des quatre reines¹?

Solution : Il y a deux solutions :





C2. On souhaite afficher sur la console l'échiquier avec les reines comme sur la figure 2. La commande `print_string "\u{2655}"` permet d'imprimer le symbole UTF-8 de la reine d'un jeu d'échec.

- Écrire une fonction dont le signature est `show : int list -> unit` qui affiche sur la console l'échiquier comme sur les figures 2 et 3. **On utilisera la fonction `iter` du module `List`.**
- Écrire une fonction **récursive** dont le signature est `rec_show : int list -> unit` qui affiche sur la console l'échiquier comme sur les figures 2 et 3.

C3. On souhaite placer une reine sur la ligne r et la colonne c . D'autres reines sont déjà présentes sur l'échiquier.

- Écrire une fonction de signature `same_col : int -> int list -> bool` qui teste si une reine située sur la colonne c est attaquée par une autre reine présente sur la même colonne. **On écrira une version récursive et une autre version qui utilise la fonction `mem` module `List`.**

1. On peut vérifier le résultat en comparant avec la séquence [A000170](#)

FIGURE 2 – Résultat sur la console de la fonction `show board` avec `board=[0;2;1;3]`FIGURE 3 – Résultat de la fonction `show` sur la liste `[3;-1;2;-1]`

- (b) Écrire une fonction de signature `same_row : int -> int list -> bool` qui teste si une reine située sur la ligne r est attaquée par une autre reine présente sur la même ligne. Si elle est attaquée, cela signifie que l'élément r de la liste est un nombre différent de -1 . **On écrira une version récursive et une autre version qui utilise la fonction `nth` du module `List`.**
- (c) Écrire une fonction de signature `down_diag : int -> int -> int list -> bool` qui teste si une reine située en (r, c) (donnés en paramètres) est attaquée par une autre reine présente sur la diagonale allant du haut vers le bas de l'échiquier. **On écrira une version récursive et une autre version qui utilise la fonction `mapi` du module `List`.**
- (d) Écrire une fonction de signature `up_diag : int -> int -> int list -> bool` qui teste si une reine située en (r, c) (donnés en paramètres) est attaquée par une autre reine présente sur la diagonale allant du bas vers le haut. **On écrira une version récursive et une autre version qui utilise la fonction `mapi` du module `List`.**
- C4. Écrire une fonction de signature `under_attack : int -> int -> int list -> bool` qui vérifie si la reine en (r, c) est attaquée par une autre reine présente sur l'échiquier. Il est nécessaire de masquer la présence éventuelle de cette reine avant de tester l'échiquier².
- C5. Écrire une fonction de signature `valid_solution : int list -> bool` teste si une configuration (liste) est une configuration valide (c'est à dire aucune reine n'est attaquée).
- C6. Résoudre le problème des quatre reines en écrivant un code qui trouve toutes les solutions par la force brute.
- Pour améliorer la compacité du code, on peut remarquer qu'avec la représentation sous forme de liste de l'échiquier, une configuration valide est forcément une permutation de la liste `[0;1;2;3;4;5;6;7]`. On peut donc se contenter de tester toutes les permutations de cette liste. On se propose donc d'écrire un code qui génère toutes les permutation d'une liste.
- C7. (a) Écrire une fonction récursive de signature `rm : 'a -> 'a list -> 'a list` qui supprime toutes les occurrences d'un élément spécifié d'une liste et renvoie la liste ainsi modifiée.

2. une sous fonction serait adaptée à la création de cette liste masquée.

- (b) Écrire une fonction de signature `create_board : int -> int list` qui renvoie la liste des entiers des 0 à $n - 1$.
- (c) Écrire une fonction récursive de signature `permutations : 'a list -> 'a list list` qui génère la liste de toutes les permutations d'une liste. On pourra raisonner comme suit :
1. si `my_list` est vide, renvoyer une liste vide,
 2. si `my_list` possède un seul élément, renvoyer la liste qui contient cet élément,
 3. sinon pour chaque élément de `my_list`, créer toutes les permutations de `my_list` sans cet élément et insérer l'élément en tête de ces listes. Concaténer toutes les listes obtenues ainsi.
- C8. Écrire une fonction de signature `brute_force_permutation : int -> unit` qui teste la validité sur l'échiquier de toutes les permutations d'une liste à n éléments. Cette fonction affiche les échiquiers solutions et le nombre de solutions sur la console.
- C9. Cette fonction est-elle efficace pour huit reines? Peut-on résoudre le problème pour des n plus grands que 8?

Solution : Pour $n = 9$, on dépasse déjà la capacité de la pile d'exécution. (Stack overflow)

Solution :

Code 1 – Résolution par la force brute du problème des n reines

```

1 let n = 4;;
2
3 let test_board = [2;0;3;2];;
4
5 let show board =
6   let print_row v =
7     for i = 0 to (List.length board) - 1 do
8       print_string (if i=v then "\u{2655} " else ". ");
9     done;
10    print_newline()
11  in List.iter print_row board; print_newline();;
12 show test_board;;
13
14 let rec_show board =
15   let print_row col =
16     for c = 0 to (List.length board) - 1 do
17       if col = c then print_string "\u{2655} " else print_string ". " done
18       ; print_newline()
19   in let rec aux b = match b with
20     | [] -> print_newline()
21     | col::t -> print_row col; aux t;
22   in aux board;;
23 rec_show test_board;;
24
25 show [3;-1;2;-1];;
26
27 let same_col c board = List.mem c board;;
28 same_col 1 test_board;;

```

```
28 same_col 3 test_board;;
29
30 let rec rec_same_col c board = match board with
31   | [] -> false
32   | head::tail -> if c = head then true else rec_same_col c tail;;
33 rec_same_col 1 test_board;;
34 rec_same_col 3 test_board;;
35
36
37 let same_row r board = (List.nth board r) != -1;;
38 same_row 1 test_board;;
39 same_row 2 test_board;;
40 same_row 3 test_board;;
41 same_row 0 test_board;;
42 same_row 1 [3;-1;2;-1];;
43 same_row 2 [3;-1;2;-1];;
44 same_row 3 [3;-1;2;-1];;
45
46 let rec_same_row r board =
47   let rec aux b i = match b with
48     | [] -> false
49     | head::tail -> if i = r && head != -1 then true else aux tail (i + 1)
50   in aux board 0;;
51 show test_board;;
52 rec_same_row 1 test_board;;
53 rec_same_row 2 test_board;;
54 rec_same_row 3 test_board;;
55 rec_same_row 0 test_board;;
56 rec_same_row 1 [3;-1;2;-1];;
57 rec_same_row 2 [3;-1;2;-1];;
58 rec_same_row 3 [3;-1;2;-1];;
59
60
61 let up_diag r c board =
62   let rc = r + c
63   in let diag board = List.mapi (fun index elem -> if elem != -1 then Some (
64     index + elem) else None) board
65   in List.mem (Some rc) (diag board);;
66 up_diag 0 0 test_board;;
67 up_diag 3 0 test_board;;
68 up_diag 0 3 test_board;;
69 up_diag 3 3 test_board;;
70 up_diag 2 2 test_board;;
71 up_diag 0 1 test_board;;
72 up_diag 1 1 test_board;;
73
74 let rec_up_diag r c board =
75   let rc = r + c
76   in let rec diag b i = match b with
77     | [] -> []
78     | head::tail -> if head != -1 then (i + head)::(diag tail (i + 1)) else
79       diag tail (i + 1)
80   in List.mem rc (diag board 0);;
81 rec_up_diag 0 0 test_board;;
82 rec_up_diag 3 0 test_board;;
```

```

81 rec_up_diag 0 3 test_board;;
82 rec_up_diag 3 3 test_board;;
83 rec_up_diag 2 2 test_board;;
84 rec_up_diag 0 1 test_board;;
85 rec_up_diag 1 1 test_board;;
86
87
88 let down_diag r c board =
89   let rc = r - c
90   in let diag board = List.mapi (fun index elem -> if elem != -1 then Some (
      index - elem) else None) board
91   in List.mem (Some rc) (diag board);;
92 down_diag 0 0 test_board;;
93 down_diag 3 0 test_board;;
94 down_diag 0 3 test_board;;
95 down_diag 3 3 test_board;;
96 down_diag 2 2 test_board;;
97 down_diag 0 1 test_board;;
98 down_diag 1 1 test_board;;
99 down_diag 2 1 test_board;;
100
101 let rec_down_diag r c board =
102   let rc = r - c
103   in let rec diag b row = match b with
104     | [] -> []
105     | col::tail -> if col != -1 then (row - col)::(diag tail (row + 1))
      else diag tail (row + 1)
106   in List.mem rc (diag board 0);;
107 rec_down_diag 0 0 test_board;;
108 rec_down_diag 3 0 test_board;;
109 rec_down_diag 0 3 test_board;;
110 rec_down_diag 3 3 test_board;;
111 rec_down_diag 2 2 test_board;;
112 rec_down_diag 0 1 test_board;;
113 rec_down_diag 1 1 test_board;;
114 rec_down_diag 2 1 test_board;;
115
116
117
118 let attack_test_board = [2;-1;3;-1];;
119 rec_show attack_test_board;;
120 show attack_test_board;;
121 let under_attack r c board =
122   let masked = List.mapi (fun index elem -> if r = index && c = elem then -1
      else elem) board
123   in same_row r masked || same_col c masked || up_diag r c masked ||
      down_diag r c masked;;
124 under_attack 0 0 attack_test_board;;
125 under_attack 1 0 attack_test_board;;
126 under_attack 1 1 attack_test_board;;
127 under_attack 1 2 attack_test_board;;
128 under_attack 1 3 attack_test_board;;
129 under_attack 0 2 attack_test_board;;
130 under_attack 3 1 attack_test_board;;
131 same_row 0 attack_test_board;;

```

```

132 same_row 1 attack_test_board;;
133 same_row 2 attack_test_board;;
134 same_row 3 attack_test_board;;
135 same_col 0 attack_test_board;;
136 same_col 1 attack_test_board;;
137 same_col 2 attack_test_board;;
138 same_col 3 attack_test_board;;
139 down_diag 3 0 attack_test_board;;
140 up_diag 3 0 attack_test_board;;
141 down_diag 3 1 attack_test_board;;
142 up_diag 3 1 attack_test_board;;
143
144
145 let rec_under_attack r c board =
146   let masked =
147     let rec aux b row = match b with
148       | [] -> []
149       | col::tail -> if (row,col) = (r,c) then -1::(aux tail (row + 1)) else
150                       col::(aux tail (row + 1))
151     in aux board 0
152   in same_row r masked || same_col c masked || up_diag r c masked || down_diag
153     r c masked;;
154 rec_under_attack 0 0 attack_test_board;;
155 rec_under_attack 1 0 attack_test_board;;
156 rec_under_attack 1 1 attack_test_board;;
157 rec_under_attack 1 2 attack_test_board;;
158 rec_under_attack 1 3 attack_test_board;;
159 rec_under_attack 0 2 attack_test_board;;
160 rec_under_attack 3 1 attack_test_board;;
161
162 let valid_board = [2;0;3;1];;
163
164 let valid_solution board =
165   let res = List.mapi (fun r c -> under_attack r c board) board
166   in not (List.mem true res);;
167 valid_solution attack_test_board;;
168 valid_solution test_board;;
169 valid_solution valid_board;;
170
171 let rec_valid_solution board =
172   let rec check row b = match b with
173     | [] -> true
174     | col::tail -> if under_attack row col board then false else check (row
175       + 1) tail
176   in check 0 board;;
177 rec_valid_solution attack_test_board;;
178 rec_valid_solution test_board;;
179 rec_valid_solution valid_board;;
180
181 let raw_force_4_queens () =
182   let board = []
183   in for i=0 to n - 1 do
184     let bi = i::board in

```

```

184         for j=0 to n - 1 do
185             let bij = j::bi in
186             for k=0 to n - 1 do
187                 let bijk = k::bij in
188                 for l=0 to n - 1 do
189                     let bijkl = l::bijk in
190                     if valid_solution bijkl then (Printf.printf "[%i,%i,%i,%i]\n"
191                                             i j k l; rec_show bijkl;)
192                     done;
193                 done;
194             done;
195         done;;
196 raw_force_4_queens();;
197
198
199 let rec rm x my_list = match my_list with
200 | [] -> []
201 | h::t -> if h=x then rm x t else h::rm x t;;
202
203 let rm x my_list = List.filter ((<>) x) my_list;;
204
205 let create_board s =
206     let rec aux l i = if s=i then l else i::(aux l (i + 1))
207     in aux [] 0;;
208 create_board 5;;
209
210 let create_board s = List.init s (fun x -> x);;
211 create_board 5;;
212
213 (* Fixed-head solution*)
214 let rec permutations my_list = match my_list with
215 | [] -> []
216 | x::[] -> [[x]]
217 | l -> List.fold_left (fun acc x -> acc @ List.map (fun p -> x::p) (
218     permutations (rm x l))) [] l;;
219
220 permutations (create_board 4);;
221 permutations (create_board 5);;
222 permutations (create_board 6);;
223
224 (* Fixed-head solution*)
225 let rec permutations my_list = match my_list with
226 | [] -> []
227 | x::[] -> [[x]]
228 | l -> List.fold_left (fun acc x -> acc @ List.map (fun p -> x::p) (
229     permutations (rm x l))) [] l;;
230
231 permutations (create_board 4);;
232 permutations (create_board 5);;
233 permutations (create_board 6);;
234
235 let brute_force_permutation s =

```



```

236   let sol_nb = ref 0
237   in let print_if_valid b = if valid_solution b then (incr sol_nb;
    print_string "Solution #"; print_int !sol_nb; print_newline(); rec_show
    b)
238   in let perms = permutations (create_board s)
239   in List.iter print_if_valid perms;;
240
241 brute_force_permutation 4;;
242 brute_force_permutation 5;;
243 brute_force_permutation 6;;
244 brute_force_permutation 7;;
245 brute_force_permutation 8;;
246 brute_force_permutation 9;;
247 brute_force_permutation 12;;

```

D Résolution par retour sur trace

L'algorithme de retour sur trace 1 construit au fur et à mesure les solutions partielles du problème et les rejette dès qu'il découvre une impossibilité.

Algorithme 1 Algorithme de retour sur trace

1: Fonction RETOUR_SUR_TRACE(v) 2: si v est une feuille alors 3: renvoyer Vrai 4: sinon 5: pour chaque fils u de v répéter 6: si u peut compléter une solution partielle au problème \mathcal{P} alors 7: RETOUR_SUR_TRACE(u) 8: renvoyer Faux	$\triangleright v$ est un nœud de l'arbre de recherche
---	--

On utilise la modélisation de l'échiquier précédente, c'est à dire qu'on construit la liste des colonnes occupées. On procède par ligne en positionnant d'abord une reine puis une autre sur la deuxième ligne. ... Ainsi de suite, la liste augmente de taille jusqu'à atteindre la taille n .

On n'a plus besoin de traiter le cas où la colonne est vide³ car à chaque fois qu'on envisage une solution partielle (une liste partielle), les lignes qui précèdent possèdent nécessairement une reine sur une des colonnes de l'échiquier.

D1. Comment coder « v est une feuille» en OCaml?

Solution : Lorsque la taille de la liste qui représente l'échiquier dépasse n , alors on a rempli l'échiquier. Si l'indice de la ligne considéré est n alors on est arrivé sur une feuille.

D2. Comment coder « u peut compléter une solution partielle au problème \mathcal{P} »?

3. que l'on avait dû traiter dans le cas de la force brute avec -1




FIGURE 4 – Exemple d'arbre de recherche structurant l'algorithme de retour sur trace. Application au problème de quatre reines.

Solution : On considère la nouvelle position u (row, col). Alors, il faut considérer la solution si cette reine n'est pas attaquée.

```
1         if not (under_attack row col board) then
```

- D3. Implémenter un algorithme de retour sur trace pour le problème des quatre reines.
- D4. Tester ce programme sur le problème des n reines et comparer les résultats avec l'algorithme de force brute. Vérifier que vous retrouvez les mêmes résultats.
- D5. Compiler le programme. Quelle valeur de n engendre un temps d'exécution supérieur à la minute?
- D6. Implémenter cet algorithme en Python et comparer les performances.

Solution : Sur ma machine, trois fois plus vite en compilé. OCaml plus rapide que Python non compilé.

Code 2 – N queens backtracking

```

1
2 let show_sol board nb =
3     let print_row col = for c = 0 to (List.length board) - 1 do
4         if col = c then print_string "\u{2655} " else
5             print_string ". "
6         done; print_newline() in
7     let rec aux b = match b with
8     | [] -> print_newline()
9     | col::t -> print_row col; print_newline(); aux t;
10    in print_string "Solution #"; print_int nb; print_newline(); aux board
11 ;;
12
13 (* rec version *)
14 let under_attack row col board =
15     let up_diag =
16         let rec aux i b = match b with
17         | [] -> []
18         | j::t -> (i + j)::(aux (i + 1) t)
19         in aux 0 board
20     in let down_diag =
21         let rec aux i b = match b with
22         | [] -> []
23         | j::t -> (i - j)::(aux (i + 1) t)
24         in aux 0 board
25     in List.mem col board || List.mem (row + col) up_diag || List.mem (row - col)
26         down_diag
27 ;;
28
29 (* List version *)
30 let under_attack row col board =
31     let up_diag = List.mapi (fun index elem -> index + elem) board
32     in let down_diag = List.mapi (fun index elem -> index - elem) board
33     in List.mem col board || List.mem (row + col) up_diag || List.mem (row - col)
34         down_diag
35 ;;
36
37 let n_queens n =
38     let sol_nb = ref 0
39     in let rec build_solutions row board =
40         if row = n then
41             (incr sol_nb; show_sol board !sol_nb)
42         else
43             for c = 0 to n - 1 do
44                 if not (under_attack row c board) then
45                     build_solutions (row+1) (board@[c])
46             done
47     in build_solutions 0 [];
48

```

```
49 n_queens 4;;
50
51 n_queens 5;;
52
53 n_queens 6;;
54
55 n_queens 7;;
56
57 n_queens 8;;
58
59 n_queens 9;;
60
61 n_queens 12;;
62
63 n_queens 14;;
```

Code 3 – N queens backtracking

```
1 # coding: utf8
2 # nb solution --> https://oeis.org/A000170/list
3 n = 12
4 on_board = [None for _ in range(n)]
5 lower_diag = []
6 upper_diag = []
7 solutions_number = 0
8
9 def print_board():
10     for row in on_board:
11         for i in range(len(on_board)):
12             if i == row:
13                 print('\u2655', end=" ") # Queen symbol utf-8
14             else:
15                 print('.', end=" ")
16         print()
17     print("-----")
18
19
20 def under_attack(r, c):
21     return c in on_board or r + c in lower_diag or r - c in upper_diag
22
23 def bt_n_queens(r=0):
24     global solutions_number # be explicit since is modified
25     if r == n:
26         return True
27     else:
28         for c in range(n):
29             #print(r,c)
30             if not under_attack(r, c):
31                 on_board[r] = c # Set Queen
32                 lower_diag.append(r + c) # danger aware !
33                 upper_diag.append(r - c) # danger aware !
34                 if bt_n_queens(r + 1):
35                     solutions_number += 1
36                     print("New solution !", solutions_number)
37                     print_board()
```

```
38         on_board[r] = None        # Unset Queen
39         lower_diag.pop()          # remove danger
40         upper_diag.pop()          # remove danger
41     return False
42
43
44 if __name__ == "__main__":
45     print_board()
46     bt_n_queens()
```
